In[•]:=
```
<< KnotTheory`
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

In[•]:=
```
KS[PD[Loop[_]]] = 0;
KS[pd_PD] := KS[pd] = Module[{spd, a, s = 0, c, cs, A, es},
   spd = (Times @@ pd) /. X[i_, j_, k_, l_] :> If[j - l == 1 || l - j > 1, Xp, Xm][i, j, k, l];
   cs = spd /. {Xp[i_, j_, k_, l_] :> a[j, ++s, i] a[k, ++s, -j] a[-l, ++s, -k] a[-i, ++s, l],
       Xm[i_, j_, k_, l_] :> a[-j, ++s, i] a[k, ++s, j] a[l, ++s, -k] a[-i, ++s, -l]} //.
     a[i_, x__, j_] a[j_, y__, k_] :> a[i, x, y, k] /. a[i_, x__, j_] :> a[x];
   A = Table[0, {Length[cs]}, {Length[cs]}];
   Do[indices = Position[cs, #][[1, 1]] & /@ (4 i - 4 + {1, 2, 3, 4});
    A[[indices, indices]] +=
     If[Head[spd[[i]]] === Xp, {{0, 0, 0, 0}, {1, -1, 0, 0}, {0, -1, 0, 1}, {-1, 2, 0, -1}},
      {{1, -1, 0, 0}, {0, 0, 0, 0}, {-2, 1, 1, 0}, {1, 0, -1, 0}}], {i, Length[spd]}];
   es = Re[Eigenvalues[N[A + Transpose[A], 40]]] /. x_Real /; Abs[x] < 10^-20 → 0;
   -Plus @@ Sign /@ es]
KS[K_] := KS[PD[K]]
```

In[•]:=
```
Total[(KS[#] == KnotSignature[#]) & /@ AllKnots[]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

Out[•]= 802 True

In[•]:=
```
Clear[KS1];
KS1[PD[Loop[_]]] = 0;
KS1[pd_PD] := KS1[pd] = Module[{spd, a, s = 0, c, cs, A},
   spd = (Times @@ pd) /. X[i_, j_, k_, l_] :> If[j - l == 1 || l - j > 1, Xp, Xm][i, j, k, l];
   cs = spd /. {
        Xp[i_, j_, k_, l_] :> a[j, ++s, i] a[k, ++s, -j] a[-l, ++s, -k] a[-i, ++s, l],
        Xm[i_, j_, k_, l_] :> a[-j, ++s, i] a[k, ++s, j] a[l, ++s, -k] a[-i, ++s, -l]
       } //. a[i_, x__, j_] a[j_, y__, k_] :> a[i, x, y, k] /. a[i_, x__, j_] :> a[x];
   A = Table[0, {Length[cs]}, {Length[cs]}];
   Do[indices = Position[cs, #][[1, 1]] & /@ (4 i - 4 + {1, 2, 3, 4});
    A[[indices, indices]] += If[Head[spd[[i]]] === Xp, {{0, 0, 0, 0}, {1, -1, 0, 0}, {0, -1, 0, 1}, {-1, 2, 0, -1}}, {{1, -1, 0, 0}, {0, 0, 0, 0}, {-2, 1, 1, 0}, {1, 0, -1, 0}}],
    {i, Length[spd]}
   ];
   Total[If[# > 0, -1, 1] & /@ DeleteCases[Eigenvalues[A + Transpose[A]], 0]]
  ]
KS1[K_] := KS1[PD[K]]
```

In[•]:=
```
Total[(KS1[#] == KnotSignature[#]) & /@ AllKnots[]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

Out[•]= 802 True

In[•]:=
```
t = f = 0; fails1 = {};
Monitor[
  Do[If[KS1[K] == KnotSignature[K], ++t, ++f; AppendTo[fails1, K]],
   {K, AllKnots[{3, 16}]}],
  {K, t, f}
 ];
{t, f}
```

KnotTheory: Loading precomputed data in KnotTheory/12A.dts.

KnotTheory: Loading precomputed data in KnotTheory/12N.dts.

KnotTheory: Loading precomputed data in KnotTheory/13A.dts.

General: Further output of KnotTheory::loading will be suppressed during this calculation.

In[•]:=
```
{t, f, fails1}
```

Out[•]= {1701935, 0, {}}

Sample Runs

```
In[•]:= pd = PD@Knot[3, 1]
Out[•]= PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]]

In[•]:= spd = (Times @@ pd) /. X[i_, j_, k_, l_] :> If[j - l == 1 || l - j > 1, Xp, Xm][i, j, k, l]
Out[•]= Xm[1, 4, 2, 5] Xm[3, 6, 4, 1] Xm[5, 2, 6, 3]

In[•]:= s = 0;
        cs = spd /. {
              Xp[i_, j_, k_, l_] :> a[j, ++s, i] a[k, ++s, -j] a[-l, ++s, -k] a[-i, ++s, l],
              Xm[i_, j_, k_, l_] :> a[-j, ++s, i] a[k, ++s, j] a[l, ++s, -k] a[-i, ++s, -l]
             } //. a[i_, x__, j_] a[j_, y__, k_] :> a[i, x, y, k] /. a[i_, x__, j_] :> a[x]
Out[•]= a[1, 7] a[5, 11] a[9, 3] a[4, 12, 8] a[10, 2, 6]

In[•]:= MatrixForm[A = Table[0, {Length[cs]}, {Length[cs]}]]
Out[•]//MatrixForm=
⎛0 0 0 0 0⎞
⎜0 0 0 0 0⎟
⎜0 0 0 0 0⎟
⎜0 0 0 0 0⎟
⎝0 0 0 0 0⎠

In[•]:= Do[
         indices = Echo[Position[cs, #][[1, 1]] & /@ (4 i - {3, 2, 1, 0})];

         A[[indices, indices]] += If[Head[spd[[i]]] === Xp, ⎛0  0  0  0⎞
                                                           ⎜1 -1  0  0⎟
                                                           ⎜0 -1  0  1⎟
                                                           ⎝-1 2  0 -1⎠, ⎛1  -1  0  0⎞
                                                                         ⎜0   0  0  0⎟
                                                                         ⎜-2  1  1  0⎟
                                                                         ⎝1   0 -1  0⎠];

         Echo@MatrixForm[A],
         {i, Length[spd]}
        ]
```

» {1, 5, 3, 4}

» $\begin{pmatrix} 1 & 0 & 0 & 0 & -1 \\ 0 & 0 & 0 & 0 & 0 \\ -2 & 0 & 1 & 0 & 1 \\ 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$

» {2, 5, 1, 4}

» $\begin{pmatrix} 2 & -2 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & -1 \\ -2 & 0 & 1 & 0 & 1 \\ 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$

» {3, 5, 2, 4}

» $\begin{pmatrix} 2 & -2 & 0 & 0 & 0 \\ 0 & 2 & -2 & 0 & 0 \\ -2 & 0 & 2 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$

In[•]:=
```
Eigenvalues[A + Transpose[A]]
```

Out[•]= {6, 6, 0, 0, 0}

In[•]:=
```
Total[If[# > 0, -1, 1] & /@ DeleteCases[Eigenvalues[A + Transpose[A]], 0]]
```

Out[•]= −2

In[•]:=
```
pd = PD@Knot[8, 17]
```

Out[•]= PD[X[6, 2, 7, 1], X[14, 8, 15, 7], X[8, 3, 9, 4], X[2, 13, 3, 14],
X[12, 5, 13, 6], X[4, 9, 5, 10], X[16, 12, 1, 11], X[10, 16, 11, 15]]

In[•]:=
```
spd = (Times @@ pd) /. X[i_, j_, k_, l_] :> If[j - l == 1 || l - j > 1, Xp, Xm][i, j, k, l]
```

Out[•]= Xm[2, 13, 3, 14] Xm[4, 9, 5, 10] Xm[8, 3, 9, 4] Xm[12, 5, 13, 6]
Xp[6, 2, 7, 1] Xp[10, 16, 11, 15] Xp[14, 8, 15, 7] Xp[16, 12, 1, 11]

In[•]:=
```
s = 0;
cs = spd /. {
      Xp[i_, j_, k_, l_] :> a[j, ++s, i] a[k, ++s, -j] a[-l, ++s, -k] a[-i, ++s, l],
      Xm[i_, j_, k_, l_] :> a[-j, ++s, i] a[k, ++s, j] a[l, ++s, -k] a[-i, ++s, -l]
     } //. a[i_, x__, j_] a[j_, y__, k_] :> a[i, x, y, k] /. a[i_, x__, j_] :> a[x]
```

Out[•]= a[5, 11] a[32, 22] a[3, 9, 25] a[15, 1, 17] a[18, 4, 28] a[30, 16, 20]
a[7, 13, 29, 21] a[8, 24, 26, 12] a[10, 2, 14, 6] a[23, 31, 19, 27]

In[•]:=
```
MatrixForm[A = Table[0, {Length[cs]}, {Length[cs]}]]
```

Out[•]//MatrixForm=

$$\begin{pmatrix} 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0 \end{pmatrix}$$

In[•]:=
```
Do[
 Echo@i;
 indices = Echo[Position[cs, #][[1, 1]] & /@ (4 i - {3, 2, 1, 0})];
 A[[indices, indices]] += If[Head[spd[[i]]] === Xp,
   {{0, 0, 0, 0}, {1, -1, 0, 0}, {0, -1, 0, 1}, {-1, 2, 0, -1}},
   {{1, -1, 0, 0}, {0, 0, 0, 0}, {-2, 1, 1, 0}, {1, 0, -1, 0}}
  ];
 Echo@MatrixForm[A],
 {i, Length[spd]}
]
```

» 1

» {4, 9, 3, 5}

»
$$\begin{pmatrix} 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&1&-2&0&0&0&0&1&0\\ 0&0&0&1&0&0&0&0&-1&0\\ 0&0&-1&1&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0 \end{pmatrix}$$

» 2

» {1, 9, 7, 8}

»
$$\begin{pmatrix} 1&0&0&0&0&0&0&0&-1&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&1&-2&0&0&0&0&1&0\\ 0&0&0&1&0&0&0&0&-1&0\\ 0&0&-1&1&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ -2&0&0&0&0&0&1&0&1&0\\ 1&0&0&0&0&0&-1&0&0&0\\ 0&0&0&0&0&0&0&0&0&0\\ 0&0&0&0&0&0&0&0&0&0 \end{pmatrix}$$

» 3

» {3, 9, 1, 8}

» $\begin{pmatrix} 2 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & -2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & -1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -2 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$

» 4

» {7, 9, 4, 6}

» $\begin{pmatrix} 2 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & -2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2 & 0 & 0 & -2 & 0 & 0 & 0 \\ 0 & 0 & -1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 & 0 & 0 & 1 & 0 & 0 & 0 \\ -2 & 0 & 0 & 0 & 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$

» 5

» {4, 5, 10, 6}

» $\begin{pmatrix} 2 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & -2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2 & 0 & 0 & -2 & 0 & 0 & 0 \\ 0 & 0 & -1 & 2 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -2 & 2 & -1 & 1 & 0 & 0 & 0 \\ -2 & 0 & 0 & 0 & 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1 & 1 & 0 & 0 & 0 & 0 \end{pmatrix}$

» 6

» {7, 2, 10, 8}

» $\begin{pmatrix} 2 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 2 & -2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2 & 0 & 0 & -2 & 0 & 0 & 0 \\ 0 & 0 & -1 & 2 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -2 & 2 & -1 & 1 & 0 & 0 & 0 \\ -2 & 0 & 0 & 0 & 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & 2 & 1 & 0 & 0 & 0 & -2 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 & -1 & 1 & 0 & 1 & 0 & 0 \end{pmatrix}$

» 7

» {3, 8, 10, 5}

» $$\begin{pmatrix} 2 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 2 & -2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2 & 0 & 0 & -2 & 0 & 0 & 0 \\ 0 & 0 & -2 & 2 & -2 & 0 & 0 & 2 & 0 & 0 \\ 0 & 0 & 0 & -2 & 2 & -1 & 1 & 0 & 0 & 0 \\ -2 & 0 & 0 & 0 & 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & 2 & 2 & 0 & 0 & 0 & -2 & -2 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \end{pmatrix}$$

» 8

» {7, 6, 10, 2}

» $$\begin{pmatrix} 2 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -2 & 0 & 0 & 0 & 2 & 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & -2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2 & 0 & 0 & -2 & 0 & 0 & 0 \\ 0 & 0 & -2 & 2 & -2 & 0 & 0 & 2 & 0 & 0 \\ 0 & 0 & 0 & -2 & 2 & -2 & 2 & 0 & 0 & 0 \\ -2 & 0 & 0 & 0 & 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & 2 & 2 & 0 & 0 & 0 & -2 & -2 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

In[•]:=
```
Eigenvalues[A + Transpose[A]] // N
```

Out[•]= {−8.59564, 8.59564, −5.23441, 5.23441, −4.32618, 4.32618, 0., 0., 0., 0.}

In[•]:=
```
Total[If[# > 0, -1, 1] & /@ DeleteCases[Eigenvalues[A + Transpose[A]], 0]]
```

Out[•]= 0